ABSTRACT

Line Plots of individual subjects grouped by treatment for a parameter of interest can be used to observe the parameter of interest over time. The individual subject line plots are distinguished by unique symbols for each subject and the treatment associated with each subject is distinguished by different line styles. The plot also includes a double legend for subject identification and for associated treatments. A graph with such features is not directly implementable by SAS graphical procedures. A methodology including the use of the annotate feature used to implement such a graph is discussed in the paper.

INTRODUCTION

In the early phases of the clinical trial it may be advantageous to visualize an individual laboratory or efficacy parameter of interest over time for individual subjects. This allows for easier comparisons when fewer treatments and fewer number of subjects to be plotted. The graph of interest groups subjects by treatment. Each treatment is distinguished by unique line styles. The legend for the treatments is plotted using annotate within the data area. The individual subject line plots are distinguished by unique symbols for each subject and the legend for subject identifiers is plotted outside the data area and below the X-axis.

METHODOLOGY

![Illustration of graph of interest: Parameter X over time](image)

Figure 1. Illustration of graph of interest: Parameter X over time
Creating Customized Spaghetti Plots, continued

The step by step procedure to create such a graph of interest is as follows:

STEP 1

Macro variables required to identify the variable for which individual line plots are plotted, the visit/X-axis variable used and the Y-axis variable used are created.

```plaintext
%let leaf=subjid2;
%let visitn=visitnum1;
%let yvalue=logchg;
```

STEP 2

Assuming that there are two treatments, treat=1 and treat=2, for one of the treatments an extra space is inserted for the subject identifier. This ensures that when the legend is created for each of the subjects, the subjects are grouped by treatment. To distinguish the individual line plots of one treatment over the other in case of overlapping timepoints, the visit variable(visitnum1) is shifted by 0.25 for one of the treatments.

```plaintext
data srcds;
    retain visitnum1;
    set srcds;
    by &leaf visitnum;
    if first.visitnum then visitnum1=visitnum;
    if not first.visitnum then visitnum1=visitnum1+0.25;
    if treat eq 2 then subjid2=' '||subjid2;
run;
```

STEP 3

An annotate dataset is created for the treatment legend. Different line styles are used to distinguish each of the treatments.

```plaintext
data annotl;
    length function $8 text $100;
    retain xsys ysys '2' when 'a' size 0.8;
    function='label';  text="-- -- -- Placebo";  x=13;y=0.9;color='black';size=1;
    output;
    function='label';  text="______ Drug X";   x=13.5;y=0.78;color='black';size=1;
    output;
    function='move'; x=11.5; y=0.63; output;
    function='draw'; x=15; y=0.63; line=1; output;
    function='draw'; x=15; y=0.95; line=1; output;
    function='draw'; x=11.5; y=0.95; line=1; output;
    function='draw'; x=11.5; y=0.63; line=1; output;
run;
```

STEP 4

To distinguish treatments by different line styles in the figure two annotate datasets are created. The two annotate datasets are created based only on ONE of the treatments. The first annotate dataset connects the data points of the treatment with a white line. The idea is to overwrite the line drawn by the SYMBOL statement. Note that the color used is white and the value of ‘When’ in the annotate dataset is specified as ‘a’ for after so that the annotate dataset is implemented ‘after’ the GPLOT statements are executed.
Creating Customized Spaghetti Plots, continued

```sas
proc sort data=srcds nodupkey out=inds2;
   by &leaf &visitn ;
   where treat=2 and not missing(&yvalue);
run;

data anno1;
   length function $8 ;
   retain  line 1 xsys ysys '2' line 21 color 'black' when 'a' size 1;
   set inds2;
   by &leaf;
   if first.&leaf then do;
      function='move'; x=&visitn;y=&yvalue;line=1;color='white';
      output;
   end;
   if not first.&leaf then do;
      function='draw';x=&visitn;y=&yvalue;line=1;color='white';
      output;
   end;
run;

STEP 5

The second annotate dataset connects the data points of the same treatment with the appropriate line style (e.g. a dashed line style). Both the annotate datasets are then combined together to create the final annotate dataset.

```sas
data anno2;
   length function $8 ;
   retain  line 1 xsys ysys '2' line 21 color 'black' when 'a' size 1;
   set inds2;
   by &leaf;
   if first.&leaf then do;
      function='move'; x=&visitn;y=&yvalue;line=21;color='black';
      output;
   end;
   if not first.&leaf then do;
      function='draw';x=&visitn;y=&yvalue;line=21;color='black';
      output;
   end;
run;

data annof;
   set anno1 anno2 annot1;
run;

STEP 6

The horizontal and vertical axis statements are then defined. The legend along with the symbols used to identify the subjects is also defined.

```sas
axis1 offset=(1cm,2cm) split="~" label=(h=0.8 font='arial' "Study Day")
value=(font='arial' h=0.75 angle=0 rotate=0) minor=none major=none ;

axis2 label=(font='arial' h=0.8 angle=90 "Parameter X") value=(font='arial' h=0.75)
minor=none ;

legend1 label=none value=(h=.8 justify=left) position=(outside bottom center) frame;
```
Creating Customized Spaghetti Plots, continued

```sas
symbol1 v=dot c=black i=join mode=include l=1;
symbol2 v=circle c=black i=join mode=include l=1;
symbol3 v=triangle c=black i=join mode=include l=1;
symbol4 v=star c=black i=join mode=include l=1;
symbol5 font=special v=L i=join c=black mode=include l=1;
symbol6 font=special v=K i=join c=black mode=include l=1;
symbol7 v== c=black i=join mode=include l=1;
symbol8 v== c=red i=join mode=include l=1;
symbol9 v== c=blue i=join mode=include l=1;
symbol10 v== c=green i=join mode=include l=1;
```

**STEP 7**

The dataset is then plotted using the GLOT procedure.

```sas
proc gplot data=srcds;
   plot &yvalue*visitnum1=&leaf/anno=annof
      haxis=axis1 vaxis=axis2 legend=legend1;
run;
quit;
```

**CONCLUSION**

A methodology to create a customized spaghettii plot has been shown. The same principles used to segregate subjects by treatment can be used to create similar plots for more than two treatments.

**ACKNOWLEDGEMENT**

I would like to thank my colleague Marek Solak for suggesting to insert a white space for subjects of one treatment over the other so that they would appear in order of treatment in the legend.

**CONTACT INFORMATION**

Your comments and questions are valued and encouraged. Please contact the author at:

Name: Sanjiv Ramalingam  
Enterprise: Vertex Pharmaceuticals, Inc.  
Address: 130 Waverly Street  
City, State, ZIP: Cambridge, MA 02139  
Email: sanjiv_ramalingam@vrtx.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies.